Mplus code corresponding to models 1 through 5 in Preacher, Rucker, and Hayes (2007)

Andrew F. Hayes The Ohio State University 9 August 2011

Preacher, K. J., Rucker, D. D., & Hayes, A. F. (2007). Addressing moderated mediation hypotheses: Theory, methods, and prescriptions. *Multivariate Behavioral Research*, 42, 185-227.

```
TITLE: Preacher, Rucker, and Hayes (2007) Model 1
DATA:
      FILE IS C:\mplus3.txt;
      FORMAT is FREE;
VARIABLE:
     names are x m y;
      usevariables are x m y xm;
DEFINE:
     xm = x*m;
ANALYSIS:
      !bootstrap = 5000;
MODEL:
      y on m (b1)
          xm (b2);
     m on x (a1);
     xm with m;
MODEL CONSTRAINT:
     new (ind xmodval);
     xmodval = -1;
     ind = a1*(b1+b2*xmodval);
output:
     !cinterval (bcbootstrap);
!remove exclamation points (!) in code above;
!to generate bias corrected bootstrap confidence intervals;
!for all effects;
!Set 'xmodval' to the value of X at which you desire the;
!estimate of the conditional indirect effect of X. This;
!conditional indirect effect will be produced in the output;
!with parameter label 'ind'. As written above, the program;
!generates the conditional indirect effect of X on Y through M;
!when X = -1;
```

```
TITLE: Preacher, Rucker, and Hayes (2007) Model 2
Called a 'direct effect and first stage' model by
        Edwards and Lambert (2007)
DATA:
      FILE IS C:\mplus3.txt;
      FORMAT is FREE;
VARIABLE:
      names are x m y w;
      usevariables are x m y w xw;
DEFINE:
      xw = x*w;
ANALYSIS:
     !bootstrap = 5000;
MODEL:
      y on m (b1)
           Х
           W
           XW;
      m on x (a1)
           xw (a3);
MODEL CONSTRAINT:
      new (ind wmodval);
      wmodval = -1;
      ind=(a1+a3*wmodval)*b1;
output:
      !cinterval (bcbootstrap);
!remove exclamation points (!) in code above;
!to generate bias corrected bootstrap confidence intervals;
!for all effects;
!Set 'wmodval' to the value of W at which you desire the;
!estimate of the conditional indirect effect of X. This;
!conditional indirect effect will be produced in the output;
!with parameter label 'ind'. As written above, the program;
!generates the conditional indirect effect of X on Y through M;
!when W = -1;
```

```
TITLE: Preacher, Rucker, and Hayes (2007) Model 3
        Called a 'second stage' model by Edwards and Lambert (2007)
DATA:
      FILE IS C:\mplus3.txt;
      FORMAT is FREE;
VARIABLE:
      names are x m y w;
     usevariables are x m y w mw;
DEFINE:
     mw = m*w;
ANALYSIS:
     !bootstrap = 5000;
MODEL:
     y on m (b1)
           X
           W
          mw (b3);
      m on x (a1);
      w with m;
     mw with m;
MODEL CONSTRAINT:
     new (ind wmodval);
     wmodval = -1;
     ind=a1*(b1+b3*wmodval);
output:
      !cinterval (bcbootstrap);
!remove exclamation points (!) in code above;
!to generate bias corrected bootstrap confidence intervals;
!for all effects;
!Set 'wmodval' to the value of W at which you desire the;
!estimate of the conditional indirect effect of X. This;
!conditional indirect effect will be produced in the output;
!with parameter label 'ind'. As written above, the program;
!generates the conditional indirect effect of X on Y through M;
!when W = -1;
```

```
TITLE: Preacher, Rucker, and Hayes (2007) Model 4
      FILE IS C:\mplus3.txt;
      FORMAT is FREE;
VARIABLE:
      names are x m y w z;
      usevariables are x m y w z mz xw;
DEFINE:
     mz = m*z;
     xw = x*w;
ANALYSIS:
     !bootstrap = 5000;
MODEL:
      y on m (b1)
           Х
           W
           Z
           mz (b3)
           XW;
      m on x (a1)
           xw (a3);
      z with m;
     mz with m;
MODEL CONSTRAINT:
      new (ind wmodval zmodval);
      wmodval = 1;
      zmodval = 2;
      ind= (a1+a3*wmodval) * (b1+b3*zmodval);
output:
      !cinterval (bcbootstrap);
!remove exclamation points (!) in code above;
!to generate bias corrected bootstrap confidence intervals;
!for all effects;
!Set 'wmodval' and 'zmodval' to the values of W and Z at which you;
!desire the estimate of the conditional indirect effect of X. This;
!conditional indirect effect will be produced in the output;
!with parameter label 'ind'. As written above, the program;
!generates the conditional indirect effect of X on Y through M;
!when W = 1 and Z = 2;
```

```
TITLE: Preacher, Rucker, and Hayes (2007) Model 5
Called a 'total effect moderation model' by
        Edwards and Lambert (2007)
DATA:
      FILE IS C:\mplus3.txt;
      FORMAT is FREE;
VARIABLE:
      names are x m y w;
      usevariables are x m y w xw mw;
DEFINE:
      mw = m*w;
      xw = x*w;
ANALYSIS:
      !bootstrap = 5000;
MODEL:
      y on m (b1)
           W
           mw (b2)
           XW;
      m on x (a1)
           xw (a3);
      mw with m;
MODEL CONSTRAINT:
      new (ind wmodval);
      wmodval = -1;
      ind=(a1+a3*wmodval) * (b1+b2*wmodval);
output:
      !cinterval (bcbootstrap);
!remove exclamation points (!) in code above;
!to generate bias corrected bootstrap confidence intervals;
!for all effects;
!Set 'wmodval' to the value of W at which you desire the;
!estimate of the conditional indirect effect of X. This;
!conditional indirect effect will be produced in the output;
!with parameter label 'ind'. As written above, the program;
!generates the conditional indirect effect of X on Y through M;
!when W = -1;
```